
17 - Application Software 1.0.2.txt
new

REM This is the full install file for Project BlackBox Application Software v 1.0.2

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM Project Blackbox
REM 
REM Summary:
REM
REM Project BlackBox is the safety behind Project HeavyBlue.  
REM Project HeavyBlue is a robust computer support system that 
REM is designed to thermoelectrically assist watercooling of a 
REM high end PC, specifically a 64 bit gaming PC.  With an eye 
REM toward the future of technology, Project HeavyBlue, as well as 
REM Project BlackBox well surpass that which is necessary to  
REM adequately cool todays high end PC.
REM 
REM Project BlackBox is an RPC-210 embedded controller that
REM that utilizes an Analog Devices ADUC812 CPU operating at
REM 11.092 MHz.  Project BlackBox serves the purpose of monitoring 
REM the status of the attached PC as well as its cooling components.
REM Given a set of user definable parameters, Project BlackBox continuously
REM checks inputs against the defined limits, if the PC starts to 
REM operate outside of these limits, Project BlackBox trips Project 
REM HeavyBlue into a protective action that results in the main computer 
REM shutting down immediately, thus preventing a burn up or freeze out.
REM 
REM Consult the Project BlackBox and the Project HeavyBlue Operation
REM and Design Manuals for further information.
REM 
REM Project Start Date: October 2005
REM
REM Designer: Kevin J. Stewart
REM 
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0.0 05/08/2006 Initial Software Release
REM 1.0.1 05/18/2006 New Motherboard - New Chipset
REM 1.0.2 06/02/2006 Added OS Computer S/D Detection
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Memory Map

REM Segment 1 - Auto Start
REM Segment 2 - Main Menu
REM Segment 3 - Validation and Diagnostic Mode
REM Segment 4 - Validation and System Run Mode
REM Segment 5 - System Shutdown
REM Segment 6 - System Configuration
REM Segment 7 - First Run Program and Storage Space

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Clear RAM segments 1-7

load 1
new
save 1
load 2

new
save 2
load 3
new
save 3
load 4
new
save 4
load 5
new
save 5
load 6
new
save 6
load 7
new
save 7

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Begin Application Software Download

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the startup routine that follows BlackBox POST  
REM This sets up all Digital I/O as well as the LCD 
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 1.1 Added I/O Latch Using Analog Output 1
REM 1.2 Added First Run Detection
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

5 print "Project BlackBox Auto Start Routine"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Configure Digital I/O and LCD to 4x40 Mode
20 config line 100,2,255,0,0
25 config display 0
26 aot 1,0

REM Look for First Run Flag
27 frn = peekf(7,6000)
28 if frn > 0 then goto 200 else goto 30
30 print "I/O Configured"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Prompt User to enable Main Computer PSU
40 display(0,0,off),"Please Switch on the Main Computer PSU"
42 display(1,0,off),"Press any Key After Complete"
44 print "Waiting for user to flip PSU ON and a Keypress"
46 clear keypad
47 a = keypad(0)
50 if a > 0 then goto 60 else goto 47

Page 2



17 - Application Software 1.0.2.txt
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Prompt User to Enable Power to Initialize Digital Outputs
60 clear display
62 display(0,0,off),"Please Press the I/O Enable Button"
64 print "Please press the Digital I/O Enable Button"

REM Switch Signal is active LOW
65 a = line#(107)
66 if a = 0 then goto 70 else goto 65

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Enable I/O Latch and go to Main Menu
70 aot 1,4095
80 print "Button Pressed, I/O Initialized, Proceeding to Main Menu"
90 display(1,0,off),"Button Pressed"
100 display(2,0,off),"Loading System Menu ..."
110 execute 2
200 execute 7

save 1

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the main menu program for Project BlackBox  
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 1.1 Added System Settings
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Main Menu, Print Control Information
10 print "System Menu, Use Keypad to Choose"
15 print "Press 1 for Diagnostic Mode"
20 print "Press 2 for System Run Mode"
25 print "Press 3 for System Settings"
30 print "Press 4 for System Shutdown"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Setup Variables
40 a = 0: REM Keypad Buffer

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Output Graphics
50 clear display
55 display(0,0,off),"System Menu: Make Your Selection"
60 display(1,1,off),"1: Diagnostic Mode   2: System Run"
65 display(2,1,off),"3: System Shutdown   4: Settings"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Look For Keypress For Control Via Tight Loop
100 a = keypad(0)
105 if a = 1 then goto 200 else goto 110: REM 1 pressed on keypad
110 if a = 5 then goto 300 else goto 120: REM 2 pressed on keypad
120 if a = 9 then goto 400 else goto 125: REM 3 pressed on keypad
125 if a = 2 then goto 500 else goto 126: REM 4 pressed on keypad

126 if a = 4 then goto 127 else goto 100: REM mode pressed on keypad

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Mode Pressed, Turn OFF Fans and Main 12V Power Supply
127 line#111,off: REM Turn OFF Fans
128 line#109,off: REM Turn OFF Main 12V Power Supply
129 print "Main 12VDC Power Supply and Fans Switched OFF"
130 goto 100

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Diagnostic Mode Selected
200 print "Diagnostic Mode Selected"
205 clear display
210 display(0,0,off),"Diagnostic Mode Selected"
215 display(2,5,off),"Loading System Diagnostics ..."
220 execute 3

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM System Run Selected
300 print "System Run Mode Selected"
305 clear display
310 display(0,0,off),"System Run Mode Selected"
315 display(2,5,off),"Loading Safety Protocols ..."
320 execute 4

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM System Shutown Selected
400 print "System Shutdown Selected"
405 clear display
410 display(0,0,off),"System Shutdown Selected"
415 display(2,5,off),"Preparing For System Shutdown"
420 execute 5

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Settings Selected
500 print "System Settings Selected"
505 clear display
510 display(0,0,off),"System Settings Selected"
515 display(2,5,off),"Loading Configuration Menu"
520 execute 6

save 2

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the Project BlackBox Diagnostic Mode program  
REM Use this only for testing purposes! 
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 1.1 Added Validation Routine
REM
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM BlackBox Diagnostic Mode

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Validation
10 print "Authorization Protocols"
11 print "Input Correct Code on Keypad"
15 print 

REM Setup all variables
20 string 240,16 :REM setup string memory
25 $(0) = "147M2580369FSUDP": REM setup keypad chars
26 $(1) = "" :REM code to be validated
27 $(2) = "" :REM code to be validated
28 $(3) = "" :REM code to be validated
29 $(4) = "" :REM code to be validated
35 $(5) = peek$(7,5000): REM correct code
40 p = 1: REM position pointer
45 chk = 1: REM validation check variable
50 a = 0: REM KEYPAD BUFFER

REM Prompt for User Input
100 print "Enter 4 digit keycode for validation"
102 clear display
104 display(0,0,off),"Authorization Required"
106 display(1,0,off),"Please Enter 4 Digit Passcode:"
108 display(1,31,off),"-    -"

REM Look for Keypresses
150 if p = 5 then 250: REM 4 digit code, build string
155 a = keypad(0): REM get keypress
160 if a = 4 then goto 330 else 165: REM mode - exit to main menu
165 if a > 0 then goto 190 else 155: REM loopback

REM Start processing key presses here
190 b = str(7,$(p),$(0),a,1): REM get  character to keypress
195 display(1,31+p,off),"*": REM fill empty space on display
200 p = p + 1: REM increment counter pointer
205 goto 150: REM get more keypresses

REM Build String for Validation
250 c = str(6,$(1),$(2)): REM build validation string
255 c = str(6,$(1),$(3)): REM build validation string
260 c = str(6,$(1),$(4)): REM build validation string
265 print $(1): REM output the entered string
270 chk = str(9,$(1),$(5)): REM validate
275 if chk = 0 then goto 280 else 305: REM 0 returned if pass
280 print "Code Accepted, Access Granted"
285 clear display
300 goto 500
305 print "Invalid Passode, Access Denied"
310 clear display
315 display(0,0,off),"Invalid Passcode, Access Denied"
320 display(2,5,off),"Exiting to Main Menu"
325 execute 2
330 print "Mode Pressed - Exit to Main Menu"
335 clear display
340 display(0,0,off),"Authorization Aborted"
345 display(2,5,off),"Exiting to Main Menu"
350 execute 2

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Diagnostics Start Here
500 print "Diagnostic Mode"
505 print "For Testing and Troubleshooting Purposes Only"
510 print "Use Keypad for Controls"

515 print
520 print "Press 1 to Toggle Cpu Cooler Control"
525 print "Press 2 to Toggle Gpu Cooler Control"
530 print "Press 3 to Toggle Chipset Cooler COntrol"
535 print "Press 4 to Toggle 12V Supply Control"
540 print "Press 5 to Toggle Fan Control"
545 print "Press 6 to Test Alarm"
550 print "Press Mode to Exit to System Menu"
555 print
560 print "In this mode the computer cannot be started"
565 print "PSU Break line is held low, do not override"
570 print
575 clear display

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Setup Constants
600 a = 0: REM KEYPAD BUFFER
602 t1 = 3.776842e-9: REM constant for temp conversion
604 t2 = -1.864594e-5: REM constant for temp conversion
606 t3 = 0.047761: REM constant for temp conversion
608 t4 = -43.483205: REM constant for temp conversion
610 c1 = 0.006105: REM constant for CPU current conversion
612 c2 = 0.002442: REM constant for GPU current Conversion

REM 614 c3 = XXXX: REM constant for CHIPSET current conversion

616 f1 = 31H: REM formatting constant for all temperatures
618 f2 = 20H: REM formatting constant for current
620 f3 = 21H: REM formatting constant for + temperatures
622 alm = 0: REM dummy variable for alarm test

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Start Data Poll Here
624 e=line#(103): REM pump/flow sense
626 f=line#(106): REM coolant level
628 g=line#(114): REM cpu cooler control
630 h=line#(109): REM 12V supply control
632 i=line#(101): REM 12V supply status
634 j=line#(116): REM GPU cooler control
638 l=line#(111): REM fan control
640 m=line#(115): REM chipset cooler control
642 n=line#(104): REM 5VSB status

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Build Status Strings
REM Pump/Flow Sense
700 if e=0 then goto 705 else goto 715
705 $(8)="ON "
710 goto 750
715 $(8)="OFF"
720 goto 750

REM Coolant Level
750 if f=0 then goto 755 else goto 765
755 $(9)="OK "
760 goto 800
765 $(9)="LOW"
770 goto 800

REM CPU Cooler Status 
800 if g=0 then goto 805 else goto 815
805 $(10)="OFF"
810 goto 850

815 $(10)="ON "
820 goto 850

REM 12V Supply Control Status
850 if h=0 then goto 855 else goto 865
855 $(11)="OFF"
860 goto 900
865 $(11)="ON-"
870 goto 900

REM 12V Supply Status
900 if i=0 then goto 905 else goto 915
905 $(12)="Inactive"
910 goto 950
915 $(12)="Active  "
920 goto 950

REM GPU Cooler Status
950 if j=0 then goto 955 else goto 965
955 $(13)="OFF"
960 goto 1050
965 $(13)="ON "
970 goto 1050

REM Fan Control Status
1050 if l=0 then goto 1055 else goto 1065
1055 $(15)="OFF"
1060 goto 1100
1065 $(15)="ON "
1070 goto 1100

REM Chipset Control Status
1100 if m=0 then goto 1105 else goto 1115
1105 $(16)="OFF"
1110 goto 1150
1115 $(16)="ON "
1120 goto 1150

REM 5VSB Status
1150 if n=0 then goto 1155 else goto 1165
1155 $(17)="OFF"
1160 goto 1300
1165 $(17)="ON "
1170 goto 1300

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM A/D Data Conversion

REM CPU Temperature and Cooler Current Draw
1300 cpt = ain(0): REM get CPU temp data
1305 cpta = t1*cpt*cpt*cpt: REM build CPU temp data
1310 cptb = t2*cpt*cpt: REM build CPU temp data
1315 cptc = t3*cpt: REM build CPU temp data
1320 cptmp = cpta+cptb+cptc+t4: REM build CPU temp data
1325 z = str(10,$(0),f1,cptmp): REM build temp CPU data string
1330 cpucur = ain(1)*c1: REM build CPU current data
1335 z = str(10,$(1),f2,cpucur): REM build CPU current data string

REM CHIPSET Temperature and Cooler Current Draw
1400 chpt = ain(2): REM get CHIPSET temp data
1405 chpta = t1*chpt*chpt*chpt: REM build chipset temp data
1410 chptb = t2*chpt*chpt: REM build CHIPSET temp data

1415 chptc = t3*chpt: REM build CHIPSET temp data
1420 chptmp = chpta+chptb+chptc+t4: REM build CHIPSET temp data
1425 z = str(10,$(2),f1,chptmp): REM build CHIPSET temp data string

REM 1430 chipcur = ain(3)*c2: REM build CHIPSET current data
REM 1435 z = str(10,$(3),f2,chipcur): REM build CHIPSET current data string

REM GPU Temperature and Cooler Current Draw
1500 gpt = ain(4): REM get GPU temp data
1505 gpta = t1*gpt*gpt*gpt: REM build GPU temp data
1510 gptb = t2*gpt*gpt: REM build GPU temp data
1515 gptc = t3*gpt: REM build GPU temp data
1520 gptmp = gpta+gptb+gptc+t4: REM build GPU temp data
1525 z = str(10,$(4),f1,gptmp): REM build GPU temp data string
1530 gpcur = ain(5)*c2: REM build GPU current data
1535 z = str(10,$(5),f2,gpcur): REM build GPU current data string

REM Coolant Temperature
1600 resdat = ain(6): REM get Coolant TEMP data
1605 resdat1 = t1*resdat*resdat*resdat: REM build Coolant TEMP data
1610 resdatb = t2*resdat*resdat: REM build Coolant TEMP data
1615 resdatc = t3*resdat: REM build Coolant TEMP data
1620 restem = resdat1+resdatb+resdatc+t4: REM build Coolant TEMP data
1625 z = str(10,$(6),f3,restem): REM build Coolant TEMP data string

REM CASE TEMPERATURE
1750 ext = ain(7): REM get CASE TEMP data
1755 exta = t1*ext*ext*ext: REM build CASE TEMP data
1760 extb = t2*ext*ext: REM build CASE TEMP data
1765 extc = t3*ext: REM build CASE TEMP data
1770 extmp = exta+extb+extc+t4: REM build CASE TEMP data
1775 z = str(10,$(7),f3,extmp): REM build CASE TEMP data string

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Display Graphics
1800 display(0,0,off),"Pump "
1801 display(0,5,off),$(8): REM display pump info
1805 display(0,9,off),"Level ",$(9): REM display water level info
1810 display(0,19,off),"Temp",$(6),"C": REM display reservoir temp
1815 display(0,30,off),"Case",$(7),"C": REM display case temperature
1820 display(1,0,off),"CPU ",$(10): REM display cpu control info
1825 display(1,8,off),"Temp",$(0),"C",$(1),"A": REM display CPU info
1835 display(1,24,off),"12V ",$(11),"-",$(12): REM display 12V supply info
1845 display(2,0,off),"GPU ",$(13): REM display GPU control info
1850 display(2,8,off),"Temp",$(4),"C",$(5),"A": REM display GPU info
1860 display(2,24,off),"5VS ",$(17): REM display 5VSB info
1865 display(2,32,off),"Fans ",$(15): REM display fan control info
1870 display(3,0,off),"Chip ",$(16): REM display chip cooler control info
1875 display(3,9,off),"Temp",$(2),"C": REM display chipset info
REM 1875 display(3,9,off),"Temp",$(2),"C",$(3),"A":REM display chipset info
1885 display(3,25,off),"Diagnostic Mode"

REM Provide Control Break
1890 a = keypad(0): REM get keypress
1892 if a > 0 then goto 2005 else goto 624

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Keypad stuff for controls

REM CPU Cooler Control
2005 if a = 1 then goto 2010 else goto 2020 

2010 if g = 0 then goto 2011 else goto 2013
2011 line#114,on
2012 goto 624
2013 line#114,off
2014 goto 624

REM GPU Cooler Control
2020 if a = 5 then goto 2021 else goto 2035
2021 if j = 0 then goto 2022 else goto 2024
2022 line#116,on
2023 goto 624
2024 line#116,off
2025 goto 624

REM Chipset Cooler Control
2035 if a = 9 then goto 2036 else goto 2050
2036 if m = 0 then goto 2037 else goto 2039
2037 line#115,on
2038 goto 624
2039 line#115,off
2040 goto 624

REM 12V Supply Control
2050 if a = 2 then goto 2051 else goto 2065
2051 if h = 0 then goto 2052 else goto 2054
2052 line#109,on
2053 goto 624
2054 line#109,off
2055 goto 624

REM Fan Control
2065 if a = 6 then goto 2066 else goto 2080
2066 if l = 0 then goto 2067 else goto 2069
2067 line#111,on
2068 goto 624
2069 line#111,off
2070 goto 624

REM Alarm Test
2080 if a = 10 then goto 2081 else goto 2090
2081 if alm = 0 then goto 2082 else goto 2085
2082 aot 0,4095
2083 alm = 1
2084 goto 624
2085 aot 0,0
2086 alm = 0
2087 goto 624

REM Exit Diagnostics Mode
2090 if a = 4 then goto 2091 else goto 2097
2091 line#114,off :REM CPU Cooler Off
2092 line#116,off :REM GPU Cooler Off
2093 line#115,off :REM Chipset Cooler Off
2094 line#109,off :REM 12V Control Off
2095 line#111,off :REM Fan Control Off
2096 execute 2 :REM return to system menu
2097 goto 624

save 3

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the Project BlackBox System Run Mode program
REM
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 1.2 Added Validation Routine
REM 1.3 Added Date/Time Display Function
REM 2.0 Redesign for lower cycle time
REM 2.1 Added Data Logging
REM 2.2 Added Watchdog Timer Usage
REM 2.3 New Motherboard - New Chipset
REM 2.4 Added OS Computer S/D Detection
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Validation Routine
1 print "Authorization Protocols"
2 print "Input Correct Code on Keypad"
3 print 

REM Setup all variables
4 string 316,20: REM setup string memory
5 $(0) = "147M2580369FSUDP": REM setup keypad chars
6 $(1) = "": REM code to be validated
7 $(2) = "": REM code to be validated
8 $(3) = "": REM code to be validated
9 $(4) = "": REM code to be validated
10 $(5) = peek$(7,5000): REM correct code
11 p = 1: REM position pointer
12 chk = 1: REM validation check variable
13 a = 0: REM KEYPAD BUFFER

REM Prompt for User Input
14 print "Enter 4 digit keycode for validation"
15 clear display
16 display(0,0,off),"Authorization Required"
17 display(1,0,off),"Please Enter 4 Digit Passcode:"
18 display(1,31,off),"-    -"

REM Look for Keypresses
19 if p = 5 then 27: REM 4 digit code, build string for 
validation
20 a = keypad(0): REM get keypress
21 if a = 4 then goto 41 else 22: REM mode pressed exits to main menu
22 if a > 0 then goto 23 else 20: REM loopback

REM Start processing key presses here
23 b = str(7,$(p),$(0),a,1): REM get corresponding character to keypress
24 display(1,31+p,off),"*": REM fill empty space on display
25 p = p + 1: REM increment counter pointer
26 goto 19: REM get more keypresses

REM Build String for Validation
27 c = str(6,$(1),$(2)): REM build validation string
28 c = str(6,$(1),$(3)): REM build validation string
29 c = str(6,$(1),$(4)): REM build validation string
30 print $(1): REM output the entered string

31 chk = str(9,$(1),$(5)): REM validate
32 if chk = 0 then goto 33 else 36: REM 0 returned if validation passed
33 print "Code Accepted, Access Granted"
34 clear display
35 goto 100
36 print "Invalid Passode, Access Denied"
37 clear display
38 display(0,0,off),"Invalid Passcode, Access Denied"
39 display(2,5,off),"Exiting to Main Menu"
40 execute 2
41 print "Mode Pressed - Exit to Main Menu"
42 clear display
43 display(0,0,off),"Authorization Aborted"
44 display(2,5,off),"Exiting to Main Menu"
45 execute 2

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM System Run Mode Code Below

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
100 print "System Run Mode"
101 print "This is the BlackBox Protection Program"
102 print "Use Keypad for Controls"
103 print
104 print "Press Mode to Shutdown Main Computer and Exit to System Menu"
105 print
106 print "In this mode, ALL Controls are Automated"
107 print 
108 print

109 WDOG 2

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Setup Constants and Timers
rem 150 string 1000,40
151 kprss = 0: REM KEYPAD BUFFER
152 t1 = 3.776842e-9: REM constant for temp conversion
153 t2 = -1.864594e-5: REM constant for temp conversion
154 t3 = 0.047761: REM constant for temp conversion
155 t4 = -43.483205: REM constant for temp conversion
156 c1 = 0.006105: REM constant for CPU current conversion
157 c2 = 0.002442: REM constant for GPU current Conversion
rem c3 = XXXX: REM constant for CHIPSET current conversion
159 f1 = 31H: REM formatting constant for all temperatures
160 f2 = 21H: REM formatting constant for current
161 f3 = 21H: REM formatting constant for + temperatures
162 f4 = 31H: REM formatting constant for session runtime
163 f5 = 50H: REM formatting constant for total runtime
164 TRTM = peekf(7,5900): REM get total runtime for system
165 SRTM = 0: REM set session runtim to zero
166 clear tick(0): REM clear timer to calculate runtime

168 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Verify System Ready for Start
200 line#109,on: REM fire up main 12VDC power supply 

REM FLOW = 0 is good
201 FLOW = line#(103)
202 if FLOW = 0 then goto 203 else goto 901: REM no flow trip

REM C5VSBSNS = 1 is good

203 C5VSBSNS = line#(104)
204 if C5VSBSNS = 1 then goto 205 else goto 902: REM computer PSU fail trip

REM H20LVL = 0 is good
205 H20LVL = line#(106)
206 if H20LVL = 0 then goto 207 else goto 903: REM coolant low trip

REM IALM = 0 is good
207 IALM = line#(108)
208 if IALM = 0 then goto 209 else goto 904: REM intrusion alarm trip

REM M12VSNS = 1 is good
209 M12VSNS = line#(101)
210 if M12VSNS = 1 then goto 211 else goto 905: REM main 12VDC power supply fail 
trip

211 print "System Ready, BlackBox Preparing to Start Main Computer"
212 display(0,0,off),"System Ready"
213 display(1,5,off),"Preparing for Main Computer Start"

214 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Get Trip Points

220 CPUHIGHT = peekf(7,5050)
221 CPULOWT = peekf(7,5100)
222 CPUHIGHC = peekf(7,5150)
223 CPULOWC = peekf(7,5200)
224 GPUHIGHT = peekf(7,5250)
225 GPULOWT = peekf(7,5300)
226 GPUHIGHC = peekf(7,5350)
227 GPULOWC = peekf(7,5400)
228 CHPHIT = peekf(7,5450)
229 CHPLT = peekf(7,5500)
230 CHIPHIC = peekf(7,5550)
231 CHPLC = peekf(7,5600)
232 CHI = peekf(7,5650)

233 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Start Cooling System

250 print "Starting Cooling System"
251 display(2,5,off),"Starting Cooling System"
252 line#111,on: REM turn on fans
253 line#114,on: REM turn on CPU Cooler
254 line#115,on: REM turn on Chipset Cooler
255 line#116,on: REM turn on GPU Cooler

REM Verify cooling system online by current draw
REM Wait for ONE second while current values stabilize

256 clear tick(1)
257 dly = tick(1)
258 if dly > 1 then goto 259 else goto 257
259 cpucur = ain(1)*c1
260 REM chipcur = ain(3)*c2
261 gpcur = ain(5)*c2
262 if cpucur > CPULOWC then goto 263 else goto 906
263 if cpucur < CPUHIGHC then goto 264 else goto 906
264 REM if chipcur > CHPLC then goto 265 else goto 906

265 REM if chipcur < CHIPHIC then goto 266 else goto 906
266 if gpcur > GPULOWC then goto 267 else goto 906
267 if gpcur < GPUHIGHC then goto 268 else goto 906
268 print "Cooling System Sucessfully Started"

269 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM START COMPUTER

300 print "Starting Main Computer"
301 display(3,5,off),"Starting Main Computer"
302 line#113,on: REM allow computer PSU to fire
REM Simulate 0.25 Second Power Switch Press
303 line#112,on: REM push power switch
304 clear tick(1)
305 dly = tick(1)
306 if dly < 0.25 then goto 305 else goto 307
307 line#112,off: REM release power switch
REM Verify Computer PSU Startup By 5V Line
REM C5VDC = 1 is good
308 C5VDC = line#(105)
309 if C5VDC = 1 then goto 310 else goto 902
310 line#117,on: REM turn on Green LED
311 print "Main Computer Sucessfully Started"
312 display(3,30,off),"ONLINE"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Checks Passed, Preparing to Start
313 print "BlackBox Protection System Going Hot"
314 print
315 print "Date,Time,CPU Temperature,CPU Current,CHP Temperature",
316 print ",CHP Current,GPU Temperature,GPU Current,Coolant Temperature"
317 clear display

318 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Data Poll, Protection Loop Begins Here

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Time and Date Calculation
350 $(0) = "": REM reset string
351 a = time(0): REM get hours
352 gosub 400 : REM convert to string and put in buffer
353 a = str(6,$(0),":")
354 a = time(1) : REM get minutes
355 gosub 400
356 a = str(6,$(0),":")
357 a = time(2): REM get seconds
358 gosub 400
362 $(10) = $(0)
363 a = str(7,$(9),$(0),1,5)
364 display(0,35,off),$(9): REM Display Time
375 $(0) = "": REM reset string
376 a = date(1): REM get month
377 gosub 400: REM convert to string and put in buffer
378 a = str(6,$(0),"/")
379 a = date(2): REM get day of month
380 gosub 400
381 a = str(6,$(0),"/")

382 a = date(0): REM get last two digits of year
383 gosub 400
384 rem $(8) = $(0)
385 $(8) = $(0)
386 display(0,24,off),$(8): REM Display Date
387 goto 450: REM goto runtime calc

388 WDOG

REM Time and Date formatting subroutine
400 rem Convert number to string, add to string, put in : to make time
401 a = str(10,$(1),20h,a): REM force to 2 digits + space
rem Strip off leading space. Length of string is always 3
rem This is done by shifing string over by 1
402 for a = 1 to 3
403  asc($(1),a) = asc($(1),a+1)
404 next
rem If leading number is space, force to a 0
405 if asc($(1),1) = 32 then asc($(1),1) = 48
406 a = str(6,$(0),$(1)): REM append string
407 return

REM Make Runtime Calculations
450 NRTM = tick(0)
451 if NRTM < 360 then goto 456 else goto 452
452 SRTM = SRTM + 0.1
453 TRTM = TRTM + 0.1
454 pokef 7,5900,TRTM
455 clear tick(0)
456 z = str(10,$(20),f4,SRTM): REM build session runtime string
457 z = str(10,$(21),f5,TRTM): REM build total runtime string
458 display(0,16,off),$(20),"H": REM display session runtime
459 display(1,30,off),"RTM"
460 display(1,33,off),$(21),"H": REM display total runtime

461 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Get Tripppable Digital Input Statuses and Make Safety Decisions

REM M12VSNS = 1 is good
500 M12VSNS = line#(101)
501 if M12VSNS = 1 then goto 502 else goto 905

REM FLOW = 0 is good
502 FLOW = line#(103)
503 if FLOW = 0 then goto 504 else goto 901

REM C5VSBSNS = 1 is good
504 C5VSBSNS = line#(104)
505 if C5VSBSNS = 1 then goto 506 else goto 902

REM C5VDC = 1 is good
506 C5VDC = line#(105)
507 if C5VDC = 1 then goto 508 else goto 935

REM H20LVL = 0 is good
508 H20LVL = line#(106)
509 if H20LVL = 0 then goto 510 else goto 903

REM IALM = 0 is good
510 IALM = line#(108)
511 if IALM = 0 then goto 550 else goto 904

512 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Get Analog Input Data and Perform A/D Data Conversion
REM Clear all Variables
525 a = 0
526 b = 0
527 c = 0
528 d = 0
529 e = 0
530 f = 0
531 g = 0
532 $(0) = ""
533 $(1) = ""
534 $(2) = ""
535 $(3) = ""
536 $(4) = ""
537 $(5) = ""
538 $(5) = ""
539 $(6) = ""
540 $(7) = ""
541 h = 0

REM CPU Temperature and Cooler Current Draw
550 a = ain(0): REM fetch CPU temp data
551 cptemp = (t1*a*a*a)+(t2*a*a)+(t3*a)+t4: REM build CPU temp data
552 z = str(10,$(0),f1,cptemp): REM build temp CPU data string
553 b = ain(1)*c1: REM build CPU current data
554 z = str(10,$(1),f2,b): REM build CPU current data string

REM CHIPSET Temperature and Cooler Current Draw
560 c = ain(2): REM fetch CHIPSET temp data
561 chpt = (t1*c*c*c)+(t2*c*c)+(t3*c)+t4: REM build chipset temp data
562 z = str(10,$(2),f1,chpt): REM build CHIPSET temp data string
REM 566 d = ain(3)*c2: REM build CHIPSET current data
REM 567 z = str(10,$(3),f2,d): REM build CHIPSET current data string

REM GPU Temperature and Cooler Current Draw
570 e = ain(4): REM fetch GPU temp data
571 gpt = (t1*e*e*e)+(t2*e*e)+(t3*e)+t4: REM build GPU temp data
572 z = str(10,$(4),f1,gpt): REM build GPU temp data string
573 f = ain(5)*c2: REM build GPU current data
574 z = str(10,$(5),f2,f): REM build GPU current data string

REM Coolant Temperature
580 g = ain(6): REM fetch Coolant TEMP data
581 cool = (t1*g*g*g)+(t2*g*g)+(t3*g)+t4: REM build Coolant TEMP data
582 z = str(10,$(6),f3,cool): REM build Coolant TEMP data string

REM CASE TEMPERATURE
590 h = ain(7): REM fetch CASE TEMP data
591 ext = (t1*h*h*h)+(t2*h*h)+(t3*h)+t4: REM build CASE TEMP data
592 z = str(10,$(7),f3,ext): REM build CASE TEMP data string

593 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Analog Input Safety Decisions
600 if cptemp < CPUHIGHT then goto 601 else goto 907
601 if cptemp > CPULOWT then goto 602 else goto 909
602 if b < CPUHIGHC then goto 603 else goto 911

603 if b > CPULOWC then goto 604 else goto 913
604 if chpt < CHPHIT then goto 605 else goto 923
605 if chpt > CHPLT then goto 606 else goto 925
606 REM if d < CHIPHIC then goto 607 else goto 927
607 REM if d > CHPLC then goto 608 else goto 929
608 if gpt < GPUHIGHT then goto 609 else goto 915
609 if gpt > GPULOWT then goto 610 else goto 917
610 if f < GPUHIGHC then goto 611 else goto 919
611 if f > GPULOWC then goto 612 else goto 921
612 if cool < CHI then goto 650 else goto 931

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Display Output
650 display(0,0,off),"BlackBox Online"
654 display(1,0,off),"CPU:AMD X2 4800+"
655 display(1,16,off),$(0),"C": REM CPU Temperature
656 display(1,23,off),$(1),"A": REM CPU Current 
659 display(2,0,off),"GPU:ATI X800 Pro"
660 display(2,16,off),$(4),"C": REM GPU Temperature
661 display(2,23,off),$(5),"A": REM GPU Current 
662 display(2,30,off),"Case"
663 display(2,34,off),$(7),"C": REM Case Temperature 
664 display(3,0,off),"Chipset:Uli 1695"
665 display(3,16,off),$(2),"C": REM Chipset Temperature
666 REM display(3,23,off),$(3),"A": REM Chipset Current 
667 display(3,30,off),"H2O"
668 display(3,34,off),$(6),"C": REM Coolant Temperature

669 $(3) = "XXX"

REM Output Datalog Information
670 print $(8),",",$(10),",",$(0),",",$(1),",",$(2),",",$(3),",",$(4),
671 print ",",$(5),",",$(6)

672 WDOG

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Provide Control Break
700 pun1= keypad(0): REM get keypress
702 if pun1 > 0 then goto 750 else goto 350

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Keypad stuff for controls
750 if pun1 = 4 then goto 751 else goto 350
751 clear display
752 display(0,0,off),"System Shutdown Menu"
753 display(1,5,off),"1: Emergency Shutdown"
754 display(2,5,off),"2: Standard Shutdown"
755 print "Shutdown Menu, Use Keypad to Select, 5 Second Timeout"
756 print "1: Emergency Shutdown"
757 print "2: Standard Shutdown"

REM 5 Second Timer to Return to Protection Loop
758 clear tick(1)
759 dly = tick(1): WDOG
760 if dly < 5 then goto 761 else goto 764
761 pun2 = keypad(0)
762 if pun2 = 1 then goto 800 else goto 763: REM 1 - Emergency Shutdown
763 if pun2 = 5 then goto 850 else goto 759: REM 2 - Standard Shutdown
764 clear display
765 goto 350

REM Emergency Shutdown
800 print "Emergency Shutdown Selected"
801 line#113,off: REM Kill Computer PSU
802 line#114,off: REM Kill CPU Cooler
803 line#115,off: REM Kill Chipset Cooler
804 line#116,off: REM Kill GPU Cooler
805 line#117,off: REM Turn off Green LED
806 goto 890

REM Standard Shutdown 
REM Simulate 0.25 Second Power Switch Press
850 print "Standard Shutdown Selected"
851 line#112,on: REM push power switch
852 clear tick(1)
853 dly = tick(1)
854 if dly < 0.25 then goto 853 else goto 855
855 line#112,off: REM release power switch
856 clear display
857 display(2,5,off),"Waiting for Computer Shutdown"
858 print "Waiting for computer Shutdown"

REM Wait For Computer Shutdown
875 dly = line#(105)
876 WDOG
877 if dly = 0 then goto 880 else goto 875

REM Shutdown Cooling System
880 line#113,off: REM Kill Computer PSU
881 line#114,off: REM Kill CPU Cooler
882 line#115,off: REM Kill Chipset Cooler
883 line#116,off: REM Kill GPU Cooler
884 line#117,off: REM Turn off Green LED
890 WDOG 0
891 print "Shutdown Complete"
892 execute 2: REM Return to Main Menu

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Trip Code Here

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM General Error Code Blast on Screen
901 clear display: display(2,10,off),"Coolant Pump Failure": goto 950
902 clear display: display(2,10,off),"Computer PSU Failure": goto 950
903 clear display: display(2,10,off),"Low Coolant Level": goto 950
904 clear display: display(2,10,off),"Intrusion Alarm": goto 950
905 clear display: display(2,10,off),"Main Power Supply Failure": goto 950
906 clear display: display(2,10,off),"General Startup Failure": goto 950
907 clear display
908 display(2,10,off),"CPU High Temperature Trip": goto 950
909 clear display
910 display(2,10,off),"CPU Low Temperature Trip": goto 950
911 clear display
912 display(2,10,off),"CPU High Current Trip": goto 950
913 clear display
914 display(2,10,off),"CPU Low Current Trip": goto 950
915 clear display
916 display(2,10,off),"GPU High Temperature Trip": goto 950
917 clear display
918 display(2,10,off),"GPU Low Temperature Trip": goto 950

919 clear display
920 display(2,10,off),"GPU High Current Trip": goto 950
921 clear display
922 display(2,10,off),"GPU Low Current Trip": goto 950
923 clear display
924 display(2,10,off),"Chipset High Temperature Trip": goto 950
925 clear display
926 display(2,10,off),"Chipset Low Temperature Trip": goto 950
927 clear display
928 display(2,10,off),"Chipset High Current Trip": goto 950
929 clear display
930 display(2,10,off),"Chipset Low Current Trip": goto 950
931 clear display
932 display(2,10,off),"Coolant High Temperature Trip": goto 950

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM This code is for OS Computer Shutdown Detection
935 if line#(104) = 1 then goto 936 else goto 902
REM 5 Second Wait for User to confirm OS S/D
936 clear display
937 display(1,5,off),"Confirm OS Shutdown"
938 display(2,5,off),"Press any Key"
939 clear tick(1)
940 dly = tick(1): REM Wait timer
941 WDOG
942 if dly < 5 then goto 943 else goto 902: REM trip after 5 seconds
943 pun3 = keypad(0): REM get keypress
944 if pun3 > 0 then goto 945 else goto 940: REM any key is ok
945 clear display
946 display(1,5,off),"Shutdown Confirmed"
947 display(2,5,off),"Exiting to Main Menu"
948 goto 880: REM Exit to main menu normally

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Protective Action and Rest of Display
950 line#113,off: REM Kill Computer PSU
951 aot 0,4095: REM Activate Alarm
952 line#114,off: REM Kill CPU Cooler
953 line#115,off: REM Kill Chipset Cooler
954 line#116,off: REM Kill GPU Cooler
955 line#117,off: REM Turn off Green LED
956 WDOG 0: REM Kill Watchdog timer

REM Display Trip Graphics
960 display(0,0,off),"****************************************"
961 display(1,0,off),"*"
962 display(1,39,off),"*"
963 display(1,5,off),"BlackBox System Trip"
964 display(2,0,off),"*"
965 display(2,39,off),"*"
967 display(3,0,off),"****************************************"

REM One Minute Timer For Alarm Auto Kill
970 clear tick(1)
971 dly = tick(1): REM Delay for Alarm Kill (Seconds)
972 if dly < 60 then goto 980 else goto 973
973 aot 0,0: REM Kill Alarm

REM Keypress Send/Clear to Kill Alarm
980 kprss = keypad(0)
981 if kprss > 0 then goto 982 else goto 971

982 if kprss = 13 then goto 983 else goto 990
983 aot 0,0
984 goto 971

REM Keypress Func to Reset/Restart BlackBox
990 if kprss = 12 then goto 991 else goto 971
991 aot 0,0
992 clear display
993 execute 2

save 4

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the system shutdown routine   
REM This program reminds the user to switch off the computer PSU
REM This is nescessary as the outputs are flashed high on BlackBox POST
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

10 print "Shutdown Routine"
20 clear display
30 display(0,0,off),"Switch off Main Computer Power Supply"
32 display(1,0,off),"Press Any Key After Complete"
34 print "Waiting for user to flip PSU off and a Keypress"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Wait for keypress
40 a = keypad(0)
50 if a = 0 then goto 40 else goto 60
60 clear display
70 display(0,0,off),"Blackbox Ready to Shutdown"
80 display(1,0,off),"Please Press the Power Button"
90 print "Power Down BlackBox"
100 end

save 5

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the configuration program, the user can do a complete default load  
REM or using by a terminal, can load custom values for BlackBox Trip Values
REM
REM ********************************************************************************
REM 
REM Version History
REM 
REM 1.0 Initial Coding
REM 

Page 19



17 - Application Software 1.0.2.txt
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Data Table and Default Values
REM 7,5000 = Validator Passcode String
REM 7,5050 = CPU HIGH TEMP =====  50 C
REM 7,5100 = CPU Low TEMP ====== -10 C
REM 7,5150 = CPU HIGH CURRENT ==  23 A
REM 7,5200 = CPU LOW CURRENT ===   5 A
REM 7,5250 = GPU HIGH TEMP =====  50 C
REM 7,5300 = GPU LOW TEMP ====== -20 C
REM 7,5350 = GPU HIGH CURRENT ==   8 A
REM 7,5400 = GPU LOW CURRENT ===   2 A
REM 7,5450 = CHIP HIGH TEMP ====  50 C
REM 7,5500 = CHIP LOW TEMP ===== -10 C
REM 7,5550 = CHIP HIGH CURRENT =   8 A
REM 7,5600 = CHIP LOW CURRENT ==   2 A
REM 7,5650 = COOLANT HIGH TEMP =  60 C
REM 7,5900 = Total Runtime Hours

REM Command Syntax
REM a = peekf(7,####)
REM $(1) = peek$(7,####)
REM pokef 7,####,[Data]
REM poke$ 7,####,$(string)

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
5 clear display
10 print "Configuration Menu"
12 print "Use Keypad to select option"
30 display(0,0,off),"Configuration Menu:"
32 display(1,5,off),"1: Load Defaults"
34 display(2,5,off),"2: Customize Settings"

REM Wait for Keypress
40 kprss = keypad(0)
42 if kprss = 0 then goto 40 else goto 44
44 if kprss = 1 then goto 5510 else goto 46: REM 1 = load defaults
46 if kprss = 5 then goto 100 else goto 48: REM 5 = customize
48 if kprss = 4 then goto 6100 else goto 40: REM 4 = mode, exit to main menu

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Input Custom Values from Terminal
100 clear display
105 display(0,0,off),"Refer to Terminal"
110 print "Customize Settings"
112 print
113 string 1000,40
114 input "Enter Authorization Passcode            (4 Digits): ",$(5)
116 input "Enter CPU High Temperature Trip     (Default = 50): ",cpuhight 
118 input "Enter CPU Low Temperature Trip     (Default = -10): ",cpulowt
120 input "Enter CPU High Current Trip         (Default = 23): ",cpuhighc
122 input "Enter CPU Low Current Trip           (Default = 5): ",cpulowc
124 input "Enter GPU High Temperature Trip     (Default = 50): ",gpuhight
126 input "Enter GPU Low Temperature Trip     (Default = -20): ",gpulowt
128 input "Enter GPU High Current Trip          (Default = 8): ",gpuhighc
130 input "Enter GPU Low Current Trip           (Default = 2): ",gpulowc
132 input "Enter Chipset High Temperature Trip (Default = 50): ",chphit
134 input "Enter Chipset Low Temperature Trip (Default = -10): ",chplt
136 input "Enter Chipset High Current Trip      (Default = 8): ",chphic
138 input "Enter Chipset Low Current Trip       (Default = 2): ",chplc
140 input "Enter Coolant High Temperature Trip (Default = 60): ",chi

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

REM Poke custom settings into RAM
150 print "Poking Settings Into RAM"
200 poke$ 7,5000,$(5)
205 pokef 7,5050,CPUHIGHT
210 pokef 7,5100,CPULOWT
215 pokef 7,5150,CPUHIGHC
220 pokef 7,5200,CPULOWC
230 pokef 7,5250,GPUHIGHT
235 pokef 7,5300,GPULOWT
240 pokef 7,5350,GPUHIGHC
245 pokef 7,5400,GPULOWC
250 pokef 7,5450,CHPHIT
255 pokef 7,5500,CHPLT
260 pokef 7,5550,CHIPHIC
265 pokef 7,5600,CHPLC
270 pokef 7,5650,CHI

REM Read values and output to terminal then exit to menu
goto 5900

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Validator Passcode
5510 string 1000,40
5512 $(1) = "1234"
5514 poke$ 7,5000,$(1)

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM CPU VALUES
5520 CPUHIGHT = 50
5521 CPULOWT = -10
5522 CPUHIGHC = 23
5523 CPULOWC = 5

REM Poke values into RAM
5530 pokef 7,5050,CPUHIGHT
5531 pokef 7,5100,CPULOWT
5532 pokef 7,5150,CPUHIGHC
5533 pokef 7,5200,CPULOWC

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM GPU VALUES
5540 GPUHIGHT = 50
5541 GPULOWT = -20
5542 GPUHIGHC = 8
5543 GPULOWC = 2

REM Poke values into RAM
5550 pokef 7,5250,GPUHIGHT
5551 pokef 7,5300,GPULOWT
5552 pokef 7,5350,GPUHIGHC
5553 pokef 7,5400,GPULOWC

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM CHIP VALUES
5560 CHPHIT = 50
5561 CHPLT = -10
5562 CHPHIC = 8
5563 CHPLC = 2

REM Poke values into RAM
5570 pokef 7,5450,CHPHIT
5571 pokef 7,5500,CHPLT
5572 pokef 7,5550,CHIPHIC

5573 pokef 7,5600,CHPLC

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM COOLANT VALUES
5580 CHI = 60

REM Poke values into RAM
5590 pokef 7,5650,CHI

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

5600 print "Defaults Loaded"
5605 clear display

REM Get Current Settings
5900 $(2) = peek$(7,5000)
5901 b = peekf(7,5050)
5902 c = peekf(7,5100)
5903 d = peekf(7,5150)
5904 e = peekf(7,5200)
5905 f = peekf(7,5250)
5906 g = peekf(7,5300)
5907 h = peekf(7,5350)
5908 i = peekf(7,5400)
5909 j = peekf(7,5450)
5910 k = peekf(7,5500)
5911 l = peekf(7,5550)
5912 m = peekf(7,5600)
5913 n = peekf(7,5650)

REM Print Current Settings
6000 print "Authorization Passcode ======== ",$(2)
6005 print "CPU High Temperature Trip ===== ",b
6010 print "CPU Low Temperature Trip ====== ",c
6015 print "CPU High Current Trip ========= ",d
6020 print "CPU Low Current Trip ========== ",e
6025 print "GPU High Temperature Trip ===== ",f
6030 print "GPU Low Temperature Trip ====== ",g
6035 print "GPU High Current Trip ========= ",h
6040 print "GPU Low Current Trip ========== ",i
6045 print "Chipset High Temperature Trip = ",j
6050 print "Chipset Low Temperature Trip == ",k
6055 print "Chipset High Current Trip ===== ",l
6060 print "Chipset Low Current Trip ====== ",m
6065 print "Coolant High Temperature Trip = ",n

6100 execute 2

save 6

REM ################################################################################

new

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM
REM This is the BlackBox First Run Setup Program  
REM This program is automatically invoked on first startup and 
REM loads all default values for BlackBox Application Software
REM
REM ********************************************************************************
REM 

REM Version History
REM 
REM 1.0 Initial Coding
REM 
REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

1 print "BlackBox First Run"

REM Set Date and Time
2 print "Initialializing Real Time Clock"
3 print "See DS1689 Data File for Information"
4 lineb,0a10ah,31h
5 lineb,0a10bh,8fh
6 lineb,0a14bh,20h
7 lineb,0a14ah,0
8 print "Set date and time"
10 input "Enter year as a number between 0 and 99:",yr
11 if (yr >= 0) .and. (yr < 100) then 20
12 print "Year invalid.  Please re-enter."
13 goto 10
20 input "Enter month as a number between 1 and 12:",mo
21 if (mo > 0) .and. (mo < 13) then 30
22 print "Month invalid.  Please re-enter."
23 goto 20
30 print "Enter day of month between 1 and 31.  NOTE: Valid date"
32 print "checking is not perfomormed.  February 31 is a valid date."
33 input "Enter day of month:",da
34 if (da >= 1) .and. (da < 32) then 40
35 print "Day of month invalid.  Please re-enter."
36 goto 33
40 date yr,mo,da
50 print
51 input "Enter hours between 0 and 23 (24 hr format):",hr
52 if (hr >=0) .and. (hr < 24) then 60
53 print "Hours input not valid."
54 goto 51
60 input "Enter minutes between 0 and 59:",min
61 if (min >= 0) .and. (min < 60) then 70
62 print "Minutes input not valid."
63 goto 60
70 input "Enter seconds between 0 and 59:",sec
71 if (sec >= 0) .and. (sec < 60) then 80
72 print "Seconds input not valid."
73 goto 70
80 time hr,min,sec
85 lineb,0a10bh,0fh :rem enable clock updates now
90 print "Date and Time Set"

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
REM Load Defaults
REM Validator Passcode
100 string 1000,40
102 $(1) = "1234"
104 poke$ 7,5000,$(1)

REM CPU VALUES
110 CPUHIGHT = 50
112 CPULOWT = -10
114 CPUHIGHC = 23
116 CPULOWC = 5

REM Poke values into RAM
120 pokef 7,5050,CPUHIGHT

122 pokef 7,5100,CPULOWT
124 pokef 7,5150,CPUHIGHC
126 pokef 7,5200,CPULOWC

REM GPU VALUES
130 GPUHIGHT = 50
132 GPULOWT = -20
134 GPUHIGHC = 8
136 GPULOWC = 2

REM Poke values into RAM
140 pokef 7,5250,GPUHIGHT
142 pokef 7,5300,GPULOWT
144 pokef 7,5350,GPUHIGHC
146 pokef 7,5400,GPULOWC

REM CHIP VALUES
150 CHPHIT = 50
152 CHPLT = -10
154 CHPHIC = 8
156 CHPLC = 2

REM Poke values into RAM
160 pokef 7,5450,CHPHIT
162 pokef 7,5500,CHPLT
164 pokef 7,5550,CHIPHIC
166 pokef 7,5600,CHPLC

REM COOLANT VALUES
170 CHI = 60
172 pokef 7,5650,CHI

REM Set Runtime Counter to zero
180 TRT = 0
182 pokef 7,5900,TRT 

REM Kill First Run Flag
190 pokef 7,6000,0

REM ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
200 print "Defaults Loaded, Printing Confirmation"

REM Get Current Settings
300 $(2) = peek$(7,5000)
301 b = peekf(7,5050)
302 c = peekf(7,5100)
303 d = peekf(7,5150)
304 e = peekf(7,5200)
305 f = peekf(7,5250)
306 g = peekf(7,5300)
307 h = peekf(7,5350)
308 i = peekf(7,5400)
309 j = peekf(7,5450)
310 k = peekf(7,5500)
311 l = peekf(7,5550)
312 m = peekf(7,5600)
313 n = peekf(7,5650)

REM Print Current Settings
400 print "Authorization Passcode ======== ",$(2)
405 print "CPU High Temperature Trip ===== ",b
410 print "CPU Low Temperature Trip ====== ",c
415 print "CPU High Current Trip ========= ",d

420 print "CPU Low Current Trip ========== ",e
425 print "GPU High Temperature Trip ===== ",f
430 print "GPU Low Temperature Trip ====== ",g
435 print "GPU High Current Trip ========= ",h
440 print "GPU Low Current Trip ========== ",i
445 print "Chipset High Temperature Trip = ",j
450 print "Chipset Low Temperature Trip == ",k
455 print "Chipset High Current Trip ===== ",l
460 print "Chipset Low Current Trip ====== ",m
465 print "Coolant High Temperature Trip = ",n

500 execute 1

save 7

REM ################################################################################

REM Set First Run Flag
pokef 7,6000,1

